

Effectuer des calculs sur des données de type date

Après avoir abordé le principe de calcul sur les dates utilisé par Excel, nous vous proposons de découvrir quelques fonctions spécifiques au traitement des dates à travers quelques exemples.

Principe de calcul sur des jours

- ☐ Si le calcul porte sur des jours, procédez comme pour les autres calculs car Excel enregistre les dates sous la forme de nombres séquentiels appelés numéros de série. De ce fait, elles peuvent être ajoutées, soustraites et incluses dans d'autres calculs.
- ☐ Par défaut, sous Windows, Excel utilise le calendrier depuis 1900 (Excel pour Macintosh, le calendrier depuis 1904). Le 1er janvier 1900 correspond donc (pour Excel sous Windows) au numéro de série 1 et le 1er janvier 2005 correspond au nombre 38 353, car 38 353 jours se sont écoulés depuis le 1er janvier 1900.
- ☐ Pour utiliser une fonction spécifique de gestion de dates et d'heures, vous pouvez activer l'onglet Formules et cliquer sur le bouton DateHeure du groupe Bibliothèque de fonctions puis sur la fonction concernée afin d'utiliser l'Assistant.

ANNEE(numéro_de_série)

Renvoie l'année, un nombre entier entre 1900 et 9999.

Cette fonction permet d'isoler l'année d'une date ; exemple : la cellule A1 contient la valeur 12/12/2021, la fonction =ANNEE(A1) renvoie 2021.

AUJOURDHUI()

Renvoie la date du jour au format de date.

Cette fonction a la particularité de ne pas posséder d'arguments, il n'y a rien à écrire entre parenthèses, mais attention à ne pas oublier ces deux parenthèses.

DATE(année;mois;jour)

Renvoie un numéro de série.

*La fonction **DATE** permet la reconstitution d'une date à partir de trois valeurs : supposons qu'en A1, B1 et C1 nous ayons respectivement l'année 2021, le mois 12, le jour 20 ; la formule =DATE(A1;B1;C1) renverra 20/12/2021.*

DATEVAL(date_texte)

Convertir une date représentée sous forme de texte en numéro de série.

Sous forme de texte signifie que la date saisie 20/12/2021 n'est pas reconnue comme date par Excel mais comme donnée de type texte. Ceci arrive généralement lorsque des données ont été importées.

Les calculs

FIN.MOIS(date_départ;mois)

Renvoie le numéro de série du dernier jour du mois situé dans un intervalle exprimé en nombre de mois dans le futur ou le passé.

Cette fonction est très utile pour calculer les dates d'échéances.

Exemple : pour déterminer la date d'échéance d'une facture payable à 30 jours fin de mois en supposant que la date de facturation saisie en B2 est le 15/01/2022, la fonction s'écrit =FIN.MOIS(B2;1) et renvoie 28/02/2022.

FRACTION.ANNEE(date_début;date_fin;[base])

Renvoie la fraction de l'année représentant le nombre de jours entre la date de début et la date de fin.

Cette fonction est pratique pour calculer le nombre d'années d'ancienneté, l'âge et, plus généralement, le nombre d'années écoulées entre deux dates. La valeur obtenue est souvent une valeur décimale du type 12,4568256 (12 années et plus...).

HEURE(numéro_de_série)

Renvoie le nombre d'heures : un nombre entier entre 0 et 23.

Numéro_série correspond à un temps écrit de la manière suivante hh:mm:ss (heures:minutes:secondes) exemple : 12:25:30 renverra 12.

JOUR(numéro_de_série)

Donne le jour du mois (un nombre entier entre 1 et 31).

Même principe que la fonction ANNEE, elle isole le jour d'une date quelconque.

JOURS(date_fin;date_début)

Calcule le nombre de jours entre les deux dates.

À noter que vous obtenez le même résultat par simple soustraction. Par exemple : A1 contient 01/01/2022 et A2 contient 31/01/2022 en saisissant =A2-A1 vous obtenez 30.

JOURS360(date_début;date_fin;[méthode])

Calcule le nombre de jours séparant deux dates sur la base d'une année de 360 jours (12 mois de 30 jours).

JOURSEM(numéro_de_série;[type_retour])

Renvoie un chiffre entre 1 et 7 désignant le jour de la semaine d'une date.

MAINTENANT()

Renvoie la date du jour et l'heure de l'instant présent sous la forme d'une date et d'un temps comme par exemple 09/02/2022 11:14.

MINUTE(numéro_de_série)

Renvoie les minutes d'un temps.

*Comme la fonction HEURE, cette fonction isole les minutes indiquées dans un temps.
Exemple : 12:25:40 renvoie la valeur 25.*

MOIS(numéro_de_série)

Renvoie le mois d'une date.

MOIS.DECALER(date_départ;mois)

Renvoie une date qui représente une date spécifiée (l'argument date_départ), corrigée en plus ou en moins du nombre de mois indiqué.

NB.JOURS.OUVRES(date_début;date_fin;[jours_fériés])

Renvoie le nombre de jours ouvrés entiers compris entre deux dates.

NB.JOURS.OUVRES.INTL(date_début;date_fin;[weekend];[jours_fériés])

Renvoie le nombre de jours ouvrés entiers compris entre deux dates à l'aide de paramètres identifiant les jours du week-end et leur nombre (cf. section Calculer le nombre de jours ouvrés ou non entre deux dates).

NO.SEMAIN(numéro_de_série;[type_retour])

Renvoie le numéro de série en numéro de semaine dans l'année correspondant à la date indiquée.

NO.SEMAIN.ISO(date)

Renvoie le numéro ISO de la semaine de l'année correspondant à une date donnée. ISO correspond à la norme européenne.

SECONDE(numéro_de_série)

Renvoie les secondes d'un temps.

C'est le même principe que pour les fonctions HEURE et MINUTE.

SERIE.JOUR.OUVRE(date_départ;nb_jours;[jours_fériés])

Renvoie le numéro de série de la date avant ou après le nombre de jours ouvrés spécifiés (cf. section Calculer la date située après un nombre de jours ouvrés donné).

SERIE.JOUR.OUVRE.INTL(date_départ;nb_jours; [nb_jours_week-end];[jours_fériés])

Renvoie le numéro de série de la date avant et après un nombre spécifié de jours ouvrés en spécifiant des paramètres qui identifient et dénombrent les jours inclus dans le week-end.

Les calculs

TEMPS(heure;minute;seconde)

Recompose un temps à partir de trois valeurs numériques représentant les heures, les minutes et les secondes.

Exemple : A1 contient l'heure 14 ; A2 contient les minutes 19 ; A3 contient les secondes 40 ; la formule =TEMPS(A1;A2;A3) renverra 14:19:40.

TEMPSVAL(heure_texte)

Convertir une heure représentée sous forme de texte en numéro de série.

Même logique que la fonction DATEVAL pour les dates ; cette fonction permet la correction de données importées dans un mauvais format.

Combiner du texte avec une date

- ☐ Pour combiner dans une cellule le texte et la date contenus dans différentes cellules, vous pouvez utiliser la fonction TEXTE dont la syntaxe est :
=TEXTE(valeur;format_texte)

L'argument valeur représente une valeur numérique ou une formule dont le résultat est une valeur numérique ou bien encore une référence à une cellule contenant une valeur numérique.

L'argument format_texte représente un format de nombre sous forme de texte défini dans la zone Catégorie de la boîte de dialogue Format de cellule.

Voici un exemple d'utilisation :

	A	B	C	D	E
1		Noms	Date de naissance	Age	Jour et mois de naissance
2		Edmond BOSAPIN	21/04/1978	37 ans	= "né(e) le "&TEXTE(C2;"jj mmmm")
3		Juste COMILFO	17/05/1968	47 ans	né(e) le 17 mai
4		Honoré DE VOUVOIR	15/06/1991	24 ans	né(e) le 15 juin
5		Aude JAVEL	12/05/1956	59 ans	né(e) le 12 mai
6		Dimitri LE COURRIER	03/07/1987	28 ans	né(e) le 03 juillet
7		Edith MOITOU	12/04/1970	45 ans	né(e) le 12 avril

La fonction TEXTE a permis, à partir de la date de naissance, la transcription du jour en nombre et du mois en lettre ce qui, à partir du 17/05/1968, donne 17 mai.

Calculer la différence entre deux dates (fonction DATEDIF)

DATEDIF est l'une des fonctions "masquées" de l'application Excel ; elle n'apparaît pas dans l'Assistant fonction ni dans l'aide en ligne. Les fonctions masquées ont été introduites dans Excel pour des raisons de compatibilité avec d'autres tableurs, elles fonctionnent parfaitement mais ne font pas partie des fonctions "officielles" d'Excel.

Cette fonction s'avère très pratique dans le cas, par exemple, où vous souhaitez calculer l'ancienneté d'un employé en années et en mois.

☐ La syntaxe de la fonction est :

DATEDIF est **DATEDIF**(Date_début;Date_fin;Type).

L'argument **Type** représente la durée calculée et peut prendre les valeurs suivantes :

- "y" pour calculer la différence absolue en années (nombre entier).
- "m" pour calculer la différence absolue en mois.
- "d" pour calculer la différence absolue en jour.
- "ym" pour calculer le reliquat de mois une fois déduites toutes les années entières.
- "yd" pour calculer le reliquat de jours une fois déduites toutes les années entières.
- "md" pour calculer le reliquat de jours une fois déduits tous les mois entiers.

Voici un exemple d'utilisation :

	A	B	C
1		Date de début	30/11/2013
2		Date de fin	01/12/2015
4		Nombre de jours	=DATEDIF(C1;C2;"d")
5		Nombre de mois	=DATEDIF(C1;C2;"m")
6		Nombre d'années	=DATEDIF(C1;C2;"y")
7			

Les calculs

Voici un autre exemple qui permet de calculer l'âge d'une personne en fonction de la date du jour (fonction **=AUJOURD'HUI()**) :

	A	B	C	D	E
1		Date de naissance	25/02/1964	FORMULES	
3		Age en années	51 ans	=DATEDIF(C1;AUJOURDHUI();"y")&" ans"	
4		Nombre de mois	51 ans et 4 mois	=DATEDIF(C1;AUJOURDHUI();"y")&" ans et "&DATEDIF(C1;AUJOURDHUI();"ym")&" mois"	
5					

Calculer le nombre de jours ouvrés ou non entre deux dates

- Excel sait calculer le nombre de jours ouvrés (du lundi au vendredi) qui séparent deux dates à l'aide de la fonction **NB.JOURS.OUVRES** dont la syntaxe est :
=NB.JOURS.OUVRES(jour_début;jour_fin)

Voici un exemple d'utilisation :

	A	B	C
1		Date de début	30/11/2013
2		Date de fin	01/12/2015
4		Nombre de jours ouvrés	522
5		Formule de calcul	=NB.JOURS.OUVRES(C1;C2)
6			

Pour que cette fonction puisse tenir compte des jours fériés, vous devez y ajouter un troisième argument faisant référence à un jour férié ou à une plage de dates fériées.

- La syntaxe de cette fonction est alors :
=NB.JOURS.OUVRES(jour_début;jour_fin;jours_fériés)

Sur cet exemple, les jours fériés ont été calculés dans la plage de cellules B4 à B15.

Année		2015		Nombre de jours chômés <i>(hors week-end)</i>		9			
Jours fériés				Mois	Début	Fin	Nombre de jours ouvrés	Nbre de dimanches	
1	Premier de l'an	jeudi	01 janvier 2015	1	janvier	01/01/2015	31/01/2015	21	4
2	Pâques	dimanche	05 avril 2015	2	février	01/02/2015	28/02/2015	20	4
3	Lundi de Pâques	lundi	06 avril 2015	3	mars	01/03/2015	31/03/2015	22	5
4	Fête du travail	vendredi	01 mai 2015	4	avril	01/04/2015	30/04/2015	21	4
5	Victoire 1945	vendredi	08 mai 2015	5	mai	01/05/2015	31/05/2015	17	5
6	Ascension	jeudi	14 mai 2015	6	juin	01/06/2015	30/06/2015	22	4
7	Pentecôte	lundi	25 mai 2015	7	juillet	01/07/2015	31/07/2015	22	4
8	Fête nat.	mardi	14 juillet 2015	8	août	01/08/2015	31/08/2015	21	5
9	Assomption	samedi	15 août 2015	9	septembre	01/09/2015	30/09/2015	22	4
10	Toussaint	dimanche	01 novembre 2015	10	octobre	01/10/2015	31/10/2015	22	4
11	Armistice	mercredi	11 novembre 2015	11	novembre	01/11/2015	30/11/2015	20	5
12	Noël	vendredi	25 décembre 2015	12	décembre	01/12/2015	31/12/2015	22	4
L'année 2015 n'est pas une année BISSEXTILE						Total		252	52

✎ Pour calculer le nombre de jours entre deux dates (jours fériés, chômés... inclus), vous pouvez utiliser la fonction **JOURS** dont la syntaxe est **JOURS(date_fin;date_début)**.

Calculer la date située après un nombre de jours ouvrés donné

La fonction **SERIE.JOUR.OUVRE** vous permet de calculer une date correspondant à une date (date de début) plus ou moins le nombre de jours ouvrés spécifié. Les jours ouvrés excluent les samedi et dimanche ainsi que toutes les dates identifiées comme étant des jours chômés.

☐ La syntaxe de cette fonction est la suivante :

=SERIE.JOUR.OUVRE(date_début;nb_jours;jours_fériés) :

date_début Représente la date de début.

nb_jours Représente le nombre de jours ouvrés avant ou après la date de début. Un nombre de jours positif donne une date future, à l'inverse, un nombre de jours négatif donne une date passée.

jours_fériés Représente une liste de dates à exclusion du calendrier des jours de travail (jours fériés, congés, absence...). Cet argument est facultatif.

A. Présentation

1. Les données sources

Dirigeant d'une petite entreprise de construction de maisons à ossatures bois, vous souhaitez suivre les heures effectuées par vos employés. Trois catégories de personnels travaillent dans l'entreprise : des ouvriers, des charpentiers et éventuellement des intérimaires.

Dans la feuille **Heures** du classeur **SuiviChantiers.xlsx**, votre assistante mémorise quotidiennement les heures effectuées par chaque personne.

	A	B	C	D	E	F	G
1	Date	N° Semaine	Chantier	LIEU	Code Salarié	NB H TRAVAILLEES	NB H INTEMPERIES
2	03/01/2022	1	LA PINEDE	CHANTIER	OU-32	7,5	
3	03/01/2022	1	LA PINEDE	ATELIER	OU-28	7,5	
4	03/01/2022	1	LOTISSEMENT DU MOULIN	ATELIER	CH-06	7,5	
5	03/01/2022	1	LOTISSEMENT DU MOULIN	CHANTIER	CH-02	7,5	
6	03/01/2022	1	LOTISSEMENT DU MOULIN	ATELIER	OU-33	7,5	
7	03/01/2022	1	LOTISSEMENT DU MOULIN	ATELIER	OU-30	7,5	
8	03/01/2022	1	LOTISSEMENT DU MOULIN	ATELIER	OU-31	4,0	
9	03/01/2022	1	LA PINEDE	CHANTIER	CH-04	7,5	
10	03/01/2022	1	LOTISSEMENT DU MOULIN	ATELIER	OU-32	7,5	
11	03/01/2022	1	LA PINEDE	ATELIER	OU-29	7,5	
12	03/01/2022	1	LOTISSEMENT DU MOULIN	ATELIER	OU-33	7,0	
13	03/01/2022	1	LA PINEDE	ATELIER	OU-32	7,0	
14	03/01/2022	1	LOTISSEMENT DU MOULIN	CHANTIER	OU-31	7,0	
15	03/01/2022	1	LA PINEDE	ATELIER	CH-04	7,0	
16	03/01/2022	1	LOTISSEMENT DU MOULIN	CHANTIER	CH-02	7,0	
17	03/01/2022	1	LOTISSEMENT DU MOULIN	CHANTIER	CH-06	7,0	
18	03/01/2022	1	LOTISSEMENT DU MOULIN	CHANTIER	OU-30	7,0	

Pilotez vos informations pour optimiser la prise de décision avec Excel

Vous souhaitez connaître en temps réel :

- ▶ le nombre d'heures mensuelles travaillées sur chaque chantier,
- ▶ le nombre d'heures d'intempéries mensuelles,
- ▶ le nombre d'heures effectuées par chaque catégorie d'employé.

2. Nombre d'heures mensuelles travaillées sur chaque chantier

La feuille **StatMens** contient des tableaux dans lesquels les formules sont à insérer.

	A	B	C	D	E	F	G
1			NB HEURES ANNUELLES PREVUES				
2		DEVIS	4 560,00	11 432,00			
3		AVENANTS					
4		TOTAL	4 560,00	11 432,00			
5							
6							
7							
8			NOMBRE HEURES TRAVAILLEES			NOMBRE HEURES INTEMPERIES	
9			LA PINEDE	LOTISSEMENT DU MOULIN		LA PINEDE	LOTISSEMENT DU MOULIN
10		Janvier					
11		Février					
12		Mars					
13		Avril					
14		Mai					
15		Juin					
16		Juillet					
17		Août					
18		Septembre					
19		Octobre					
20		Novembre					
21		Décembre					
22							

Chaque jour, notre plage de données source va voir sa taille augmenter par l'ajout de nouvelles lignes. Il est judicieux dans ce cas d'utiliser, pour les calculs, des plages dynamiques dont les tailles vont correspondre au nombre exact de lignes saisies. Ceci évite de sélectionner dans les formules des plages beaucoup plus importantes afin d'être certain que les données ajoutées seront prises en compte par les formules.

La technique de définition de plages dynamiques consiste à utiliser la fonction de calcul **DECALER** pour définir une plage nommée.

La syntaxe de la fonction est détaillée ci-dessous.

=DECALER(Cellule;Nb Li;Nb Col;Hauteur;Largeur)

Cellule est la plage à partir de laquelle opérer le décalage.

Nb Li correspond au nombre de lignes vers le bas (si positif) ou vers le haut (si négatif) dont la cellule doit être décalée.

Nb Col correspond au nombre de colonnes vers la droite (si positif) ou vers la gauche (si négatif) dont la cellule doit être décalée.

Hauteur est le nombre de lignes en hauteur de la plage renvoyée.

Largeur est le nombre de colonnes en largeur de la plage renvoyée.



La fonction DECALER ne décale pas physiquement les cellules mais renvoie uniquement les coordonnées d'une plage de cellules dont les dimensions ont été modifiées et/ou pour laquelle la position a été décalée par rapport à la plage définie dans l'argument Cellule.

- ☞ Dans l'onglet **Formules** - groupe **Noms définis**, cliquez sur **Gestionnaire de noms**, ou bien utilisez le raccourci **[Ctrl] [F3]**.
- ☞ Dans la boîte de dialogue **Gestionnaire de noms**, cliquez sur le bouton **Nouveau**.
- ☞ Entrez le nom de la plage puis saisissez la formule dans la zone **Fait référence à** :
=DECALER(Heures!\$A\$2;;;NBVAL(Heures!\$A\$2:\$A\$10000);1)

La plage étant limitée à la ligne 10000, nous avons donc considéré ici qu'il n'y aura pas plus de 9999 lignes saisies cette année. Si nous prenons en compte rapidement 10 salariés réalisant 250 jours travaillés par an, cela donne 2500 lignes, ceci nous laisse une marge suffisante, même dans le cas de l'embauche de nouveaux salariés ou d'intérimaires.


Pilotez vos informations pour optimiser la prise de décision avec Excel




Lorsque vous définissez un nouveau nom, n'insérez pas d'espace dans le nom et évitez de donner un nom qui correspond à une référence de cellule ; par exemple, JAN2023 correspond à la cellule située en colonne JAN et en ligne 2023.

☞ Validez par OK.

Le nom a été rajouté dans la liste du Gestionnaire de noms.



☞ Reproduisez les étapes 2 à 4 pour créer les autres pages nommées :



☞ Cliquez sur Fermer pour terminer.

Une fois ce travail préparatoire effectué, nous pouvons insérer les différentes formules dans nos tableaux.

Ne souhaitant pas rajouter de colonnes dans notre feuille **Heures**, nous allons utiliser dans nos formules la fonction MOIS, fonction à appliquer sur la plage nommée **Dates**.

Nombre d'heures travaillées :

- ☞ Cliquez sur l'onglet **StatMens**.
- ☞ Positionnez le curseur dans la cellule **C10**.
- ☞ Saisissez la formule :

$$=SOMME((MOIS(Dates)=EQUIV(\$B10;\$B\$10:\$B\$21;0))* (Chantiers=C\$9)*(NBHTrav))$$
- ☞ Cette formule étant une formule matricielle, validez-la par **Ctrl** **⇩** **⇨**.
- ☞ Sélectionnez ensuite la plage **C10** à **D21** puis effectuez une recopie vers le bas (**Ctrl** **B**) et vers la droite (**Ctrl** **D**).
- ☞ Formatez cette zone au format **Séparateur de milliers**.
- ☞ Rajoutez les totaux en pied de tableau.

Vos premières données statistiques doivent être telles que ci-dessous :

		NOMBRE HEURES TRAVAILLEES		NOMBRE HEURES INTEMPERIES	
		LA PINEDE	LOTISSEMENT DU MOULIN	LA PINEDE	LOTISSEMENT DU MOULIN
10	Janvier	622,00	896,50		
11	Février	532,50	810,00		
12	Mars	109,50	180,00		
13	Avril	79,50	135,00		
14	Mai	689,00	1 350,00		
15	Juin	758,50	1 950,00		
16	Juillet	-	-		
17	Août	-	-		
18	Septembre	-	-		
19	Octobre	-	-		
20	Novembre	-	-		
21	Décembre	-	-		
22		2 791,00	5 321,50		

Pilotez vos informations pour optimiser la prise de décision avec Excel

Nombre d'heures d'intempéries :

- ☞ Copiez-collez la formule située en C10 dans la cellule F10.
- ☞ Dans la formule, remplacez NBHTrav par NBHIntemp :

$$=SOMME((MOIS(Dates)=EQUIV(\$B10;\$B\$10:\$B\$21;0))*(Chantiers=F\$9)*(NBHIntemp))$$
- ☞ Validez la modification par **Ctrl** **⇩** **↵**.
- ☞ Sélectionnez ensuite la plage F10 à G21 puis effectuez une recopie vers le bas (**Ctrl** **B**) suivie d'une recopie vers la droite (**Ctrl** **D**).
- ☞ Rajoutez les totaux en pied de tableau.

Vos données statistiques doivent être maintenant telles que ci-dessous :

		NB HEURES ANNUELLES PREVUES			
DEVIS	4 560,00	11 432,00			
AVENANTS					
TOTAL	4 560,00	11 432,00			
		NOMBRE HEURES TRAVAILLEES		NOMBRE HEURES INTEMPERIES	
		LA PINEDE	LOTISSEMENT DU MOULIN	LA PINEDE	LOTISSEMENT DU MOULIN
Janvier	622,00	896,50	28,00	42,00	
Février	532,50	810,00	56,00	84,00	
Mars	109,50	180,00	-	-	
Avril	79,50	135,00	-	-	
Mai	689,00	1 350,00	30,00	45,00	
Juin	758,50	1 950,00	30,00	75,00	
Juillet	-	-	-	-	
Août	-	-	-	-	
Septembre	-	-	-	-	
Octobre	-	-	-	-	
Novembre	-	-	-	-	
Décembre	-	-	-	-	
		2 791,00	5 321,50	144,00	246,00

3. Indicateurs graphiques

Il peut être intéressant d'intégrer deux petits indicateurs afin de visualiser rapidement et en temps réel la part des heures travaillées par rapport aux heures prévues.

Nous allons donc intégrer deux petites barres de progression dans deux cellules, chacune relative à un chantier. Les barres de progression devront apparaître en rouge sur fond bleu. Le pourcentage affiché et l'échelle des barres seront de 0 à 100 %. Si le nombre d'heures réellement effectuées dépasse les heures prévues, les noms des chantiers situés en ligne 9 devront apparaître en rouge.

	A	B	C	D	E	F	G
1			NB HEURES ANNUELLES PREVUES				
2		DEVIS	4 560,00	11 432,00			
3		AVENANTS					
4		TOTAL	4 560,00	11 432,00			
5							
6							
7							
8			NOMBRE HEURES TRAVAILLEES			NOMBRE HEURES INTEMPERIES	
9			LA PINEDE	LOTISSEMENT DU MOULIN		LA PINEDE	LOTISSEMENT DU MOULIN
10		Janvier	622,00	896,50		28,00	42,00
11		Février	532,50	810,00		56,00	84,00
12		Mars	109,50	180,00		-	-
13		Avril	79,50	135,00		-	-
14		Mai	689,00	1 350,00		30,00	45,00
15		Juin	758,50	1 950,00		30,00	75,00
16		Juillet	-	-		-	-
17		Août	-	-		-	-
18		Septembre	-	-		-	-
19		Octobre	-	-		-	-
20		Novembre	-	-		-	-
21		Décembre	-	-		-	-
22			2 791,00	5 321,50		144,00	246,00

La formule :

☞ En C6, saisissez la formule =SI(C4=0;0;C22/C4)

☞ Appliquez un format **Pourcentage** à la cellule ainsi qu'un remplissage **Bleu clair**.

Paramétrage de la barre de progression :

☞ Cliquez en C6.

☞ Onglet **Accueil** - groupe **Styles** - **Mise en forme conditionnelle** - **Nouvelle règle**.

☞ Paramétrez les options de la règle :

Style de mise en forme : **Barre de données**

Minimum : Nombre, Valeur 0

Maximum : Nombre, Valeur 1